
Abstract

The roundrect macros for METAPOST provide extremely configurable, ex-
tremely versatile rectangles (including rounded corners), intended primar-
ily for inclusion in documents produced by TEX and friends. The idea
was to provide a METAPOST-based replacement for the incredibly versa-
tile tcolorbox package; the macros are far from achieving that goal. But
they are nevertheless extremely useful.

1 Introduction

While TikZ and its many accompanying packages, particularly tcolorbox, are
wonderful and powerful tools, whenever using them I inevitably feel completely
lost, and I exert great effort doing comparatively simple things. Contrariwise,
thanks to my experience with the drm and dozenal packages, writing in META-
POST is quite straightforward for me. So I decided to try to write some generalized
macros to provide functionality similar to that of tcolorbox. It’s not even close
to that kind of flexibility or power, but it’s still quite useful and versatile, so I
make it available for anyone who might be interested.

This document was typeset in accordance with the docstrip utility, which
allows the automatic extraction of code and documentation from the same docu-
ment.

2 Prerequisites and Conventions

Some prerequesites for using this package are METAPOST itself (obviously). If
you’re using the package with LATEX, the gmp package would probably be helpful;
be sure to use the latex package option. Finally, the package internally calls
TEX.mp, so that is also required. All of these should be packaged in any reasonably
modern LATEX system, such as TEXLive or MikTEX.

This documentation assumes nothing about your personal TEX or METAPOST

environment. ConTEXt and the various forms of LuaTEX have METAPOST built-
in; with pdfLATEX, the author’s choice, one can use the gmp package to include the
source directly in one’s document (that’s what’s been done in this documentation)
or develop a simple script to compile them afterwards and include them in the
source via \includegraphics (probably the quickest option, since compilation is
done in advance). Here, we simply post the plain vanilla METAPOST code, and
let you work out those details however you prefer.

3 Shapes and Styles

The core of all the action is the roundrect macro; this will set up your roundedroundrect

rectangle in the plainest way possible. The first argument is the box’s height, the
second its width, and the third its name, by which you will draw it later:

roundrect(1in,2in)(rectangle);

draw rectangle;

All the corners don’t have to be rounded; we can make them square if we
want. To do things like this, we use the macro rrborderrad(), which takes arrborderrad()

single argument giving the border radius we want; that is, how rounded we want
the corners of our rectangle. Higher values will be more rounded, lower values will
be less:

rrborderrad(10pt);

roundrect(1in,2in)(rectangle);

draw rectangle;

Notice that the corners in this, with rrborderrad() set to 10pt, are much less
rounded than the previous example. The default border radius is 40pt, which is
quite rounded.

rrborderrad() provides an easy way to set the border radius of all four
corners at once; however, we can also control each corner individually, with
rrtoplftborderrad, rrbotlftborderrad, rrtoprtborderrad, and rrbotrtborderrad,rrtoplftborderrad

rrbotlftborderrad

rrtoprtborderrad

rrbotrtborderrad

which are parameters rather than macros; that is, we define them using := rather
than as an argument in parentheses:

rrtoplftborderrad := 20pt;

rrbotlftborderrad := 40pt;

rrtoprtborderrad := 10pt;

rrbotrtborderrad := 60pt;

roundrect(1in,2in)(rectangle);

draw rectangle;

As you can see, this makes it possible to create a large variety of shapes,
including the ability to arbitrarily flatten any side of the rectangle desired just by
setting the border radius of the appropriate corners to 0pt:

rrtoplftborderrad := 0pt;

rrtoprtborderrad := 0pt;

roundrect(1in,2in)(rectangle);

draw rectangle;

Here, we’ve flattened the top border by setting the top right and top left
corners’ border radii to 0pt. This ability to flatten any given side of the rectangle
makes it much easier to combine multiple rectangles into interesting forms, which
we’ll see a bit more about later.

4 Coloring the Parts

The colors of the roundrect are extremely configurable, both on the whole and
for individual parts. The background color of the roundrect is controlled by
rrinnercolor, while the border is colored by rrbordercolor().rrinnercolor

rrbordercolor

rrbordercolor(blue);

rrinnercolor := red;

roundrect(1in,2in)(rectangle);

draw rectangle;

By default, rrinnercolor is white and rrbordercolor is black. Notice that
rrbordercolor is a macro, not a parameter; that’s because each border can be
individually colored, and this macro simply does all of them at once. We’ll see
more about that later.

You can also completely suppress the border by using rrnotop, rrnobot,rrnotop

rrnobot rrnolft, and rrnort, which is particularly useful when you want to combine
rrnolft

rrnort

multiple rectangles without making an obvious border between them. You can
combine these in any way you like:

rrbordercolor(blue);

rrinnercolor := red;

rrnotop := true;

rrnobot := true;

rrborderrad(0pt);

roundrect(1in,2in)(rectangle);

draw rectangle;

Here we’ve squared all the corners to make it easier to see what’s going on.
Each border can be colored individually and separately from the others, using

the commands you’d expect:

rrtopbordercolor := blue;

rrbotbordercolor := green;

rrlftbordercolor := red;

rrrtbordercolor := black;

rrborderrad(20pt);

roundrect(1in,2in)(rectangle);

draw rectangle;

There is obviously some difficulty in determining what part of each rounded
corner should be colored how; this ability is typically more useful with a single,
flattened side, to help it blend in better when combined with other constructs:

rrbordercolor(black);

rrbotbordercolor := green;

rrinnercolor := red;

rrborderrad(20pt);

rrbotlftborderrad := 0pt;

rrbotrtborderrad := 0pt;

roundrect(1in,2in)(rectangle);

draw rectangle;

Perhaps you don’t like the border; you’d like it thicker, or drawn with a square
rather than a circular pen. You’re in luck; rrborderpen() takes the single ar-rrborderpen()

gument of the pen you’d like to draw the border with, defined like any other
METAPOST pen:

rrborderpen(pensquare scaled 3);

roundrect(1in,2in)(rectangle);

draw rectangle;

The default border pen is pencircle scaled 1.5, so this results in a square
pen rather than a circular one, twice as thick. You can also use individual pens
for each border, as expected:

rrbotlftborderrad := 0pt;

rrbotrtborderrad := 0pt;

rrbotbordercolor := green;

rrbotborderpen := pensquare yscaled 6;

roundrect(1in,2in)(rectangle);

draw rectangle;

Here we’ve flattened the bottom border, colored it green, and drawn it with
a square pen scaled on the y-axis only by 6. Clearly, there are huge possibilities
here.

5 Drop Shadows

We can also put a shadow on the boxes using rrdropshadow, a boolean valuerrdropshadow

which defaults to false:

rrdropshadow := true;

roundrect(1in,2in)(rectangle);

draw rectangle;

The drop shadow always mimics the shape of the box itself; there is presently
no way to avoid that. If for some reason you want to, you’ll have to create a
separate roundrect and place it manually.

We can control the size and direction of the drop shadow fairly easily, however,
along with its color. Its color is controlled by rrshadowcolor, which can be setrrshadowcolor

to any arbitrary METAPOST color:

rrdropshadow := true;

rrshadowcolor := blue;

roundrect(1in,2in)(rectangle);

draw rectangle;

The position of the drop shadow is governed by rrshadowx and rrshadowy,rrshadowx

rrshadowy which will shift the roundrect on the x or y axis, respectively. By default, these
are set to one quarter of the border radius in effect for the bottom left corner:

rrdropshadow := true;

rrshadowcolor := blue;

roundrect(1in,2in)(rectangle);

draw rectangle;

6 Including Text

Finally, we can put text in the rectangles; this is as configurable as everything
else:

rrbodytext := "Let’s put some text into this rectangle and see if it typesets correctly!";

roundrect(1in,2in)(rectangle);

draw rectangle;

Let’s put some text into this
rectangle and see if it typesets

correctly!

The font and style of the text can be controlled with rrtextfont, and therrtextfont

color of the text can be controlled with rrtextcolor:rrtextcolor

rrbodytext := "Text in a label";

rrtextcolor := green;

rrtextalign := "\raggedleft";
rrtextfont := "\fontsize17pt19pt\ selectfont\ itshape";

roundrect(1in,2in)(rectangle);

draw rectangle;

Text in a label

We also used, without explaining it first, rrtextalign, which allows inser-rrtextalign

tion of text alignment commands. This can also be inserted in the rrtextfont

variable, but it seemed logical to have a separate parameter for it. It’s default is
\centering.

The width of the text is governed by rrtextwd, which defaults to the samerrtextwd

width as the rectangle with a 3pt buffer on either side. The buffer is not directly
controllable, but the width can be set however you like:

rrbodytext := "Let’s put some text into this rectangle and see if it typesets correctly!";

rrtextwd := 80pt;

roundrect(1in,2in)(rectangle);

draw rectangle;

Let’s put some
text into this

rectangle and see
if it typesets

correctly!

Finally, to restore all these values to the default, use the rrrestorevals;

directive. This will clear everything to default so you can have a completely
different roundrect in the same figure.

7 Implementation

1 input TEX;

2 color rrinnercolor; rrinnercolor := white;

3 numeric rrtoprtborderrad; rrtoprtborderrad := 40pt;

4 numeric rrbotrtborderrad; rrbotrtborderrad := 40pt;

5 numeric rrbotlftborderrad; rrbotlftborderrad := 40pt;

6 numeric rrtoplftborderrad; rrtoplftborderrad := 40pt;

7 numeric rrtextwd; rrtextwd := 0;

8 numeric rrshadowx; rrshadowx := rrbotrtborderrad/4;

9 numeric rrshadowy; rrshadowy := -rrbotrtborderrad/4;

10 string rrtextfont; rrtextfont := "\fontsize{10pt}{12pt}\selectfont ";

11 color rrtextcolor; rrtextcolor := black;

12 string rrbodytext; rrbodytext := "";

13 string rrtextalign; rrtextalign := "\centering";

14 boolean rrnotop; rrnotop := false;

15 boolean rrnobot; rrnobot := false;

16 boolean rrnolft; rrnolft := false;

17 boolean rrnort; rrnort := false;

18 boolean rrdropshadow; rrdropshadow := false;

19 color rrtopbordercolor; rrtopbordercolor := black;

20 color rrbotbordercolor; rrbotbordercolor := black;

21 color rrlftbordercolor; rrlftbordercolor := black;

22 color rrrtbordercolor; rrrtbordercolor := black;

23 color rrshadowcolor; rrshadowcolor := black;

24 def rrbordercolor(expr x) =

25 rrtopbordercolor := x;

26 rrbotbordercolor := x;

27 rrlftbordercolor := x;

28 rrrtbordercolor := x;

29 enddef;

30 def rrborderrad(expr x) =

31 rrtoplftborderrad := x;

32 rrbotlftborderrad := x;

33 rrtoprtborderrad := x;

34 rrbotrtborderrad := x;

35 enddef;

36 pen rrtopborderpen; rrtopborderpen := pencircle scaled 1.5;

37 pen rrbotborderpen; rrbotborderpen := pencircle scaled 1.5;

38 pen rrlftborderpen; rrlftborderpen := pencircle scaled 1.5;

39 pen rrrtborderpen; rrrtborderpen := pencircle scaled 1.5;

40 def rrborderpen(expr x) =

41 rrtopborderpen := x;

42 rrbotborderpen := x;

43 rrlftborderpen := x;

44 rrrtborderpen := x;

45 enddef;

46 def rrrestorevals =

47 rrborderrad(40pt);

48 rrbordercolor(black);

49 rrborderpen(pencircle scaled 1.5);

50 rrinnercolor := white;

51 rrnotop := false;

52 rrnobot := false;

53 rrnolft := false;

54 rrnort := false;

55 rrtextwd := 0;

56 rrtextfont := "\fontsize{10pt}{12pt}\selectfont ";

57 rrtextcolor := black;

58 rrbodytext := "";

59 rrtextalign; rrtextalign := "\centering";

60 rrdropshadow := false;

61 rrshadowcolor := black;

62 rrshadowx := rrbotrtborderrad/4;

63 rrshadowy := -rrbotrtborderrad/4;

64 enddef;

65 def roundrect(expr rrht, rrwd)(suffix name) =

66 TEXPRE("%&latex" & char(10) & "\documentclass{article}\begin{document}");

67 TEXPOST("\end{document}");

68 if (rrtextwd = 0):

69 rrtextwd := rrwd - 12pt;

70 fi

71 path rra; path rrb; path rrc; path rrd;

72 pair a; pair b; pair c; pair d;

73 a := (0,0) shifted (-rrwd/2,-rrht/2);

74 b := (0,0) shifted (rrwd/2,-rrht/2);

75 c := (0,0) shifted (rrwd/2,rrht/2);

76 d := (0,0) shifted (-rrwd/2,rrht/2);

77 rra := fullcircle scaled rrbotlftborderrad shifted (xpart a +

78 (rrbotlftborderrad/2),ypart a + (rrbotlftborderrad/2));

79 rrb := fullcircle scaled rrbotrtborderrad shifted (xpart b -

80 (rrbotrtborderrad/2),ypart b + (rrbotrtborderrad/2));

81 rrd := fullcircle scaled rrtoplftborderrad shifted (xpart d +

82 (rrtoplftborderrad/2),ypart d - (rrtoplftborderrad/2));

83 rrc := fullcircle scaled rrtoprtborderrad shifted (xpart c -

84 (rrtoprtborderrad/2),ypart c - (rrtoprtborderrad/2));

85 pair f; f := (a--b) intersectionpoint rra;

86 pair g; g := (a--b) intersectionpoint rrb;

87 pair h; h := (b--c) intersectionpoint rrb;

88 pair i; i := (b--c) intersectionpoint rrc;

89 pair j; j := (c--d) intersectionpoint rrc;

90 pair k; k := (c--d) intersectionpoint rrd;

91 pair l; l := (d--a) intersectionpoint rrd;

92 pair m; m := (d--a) intersectionpoint rra;

93 picture name;

94 picture border;

95 picture rrtext;

96 pair n; pair o;

97 path rrtoplftcorner; path rrbotlftcorner;

98 path rrtoprtcorner; path rrbotrtcorner;

99 path rrtopborder; path rrbotborder;

100 path rrlftborder; path rrrtborder;

101 rrtoplftcorner := l{up}..{right}k;

102 rrtoprtcorner := j{right}..{down}i;

103 rrbotrtcorner := h{down}..{left}g;

104 rrbotlftcorner := f{left}..{up}m;

105 rrtopborder := rrtoplftcorner--rrtoprtcorner;

106 rrbotborder := rrbotrtcorner--rrbotlftcorner;

107 rrlftborder := rrbotlftcorner--rrtoplftcorner;

108 rrrtborder := rrtoprtcorner--rrbotrtcorner;

109 picture rrdropshadowpic;

110 if (rrdropshadow = true):

111 rrdropshadowpic := image(fill rrtoplftcorner--rrtoprtcorner--

112 rrbotrtcorner--rrbotlftcorner--cycle

113 shifted (rrshadowx,rrshadowy) withcolor

114 rrshadowcolor);

115 else:

116 rrdropshadowpic := currentpicture;

117 fi

118 name := currentpicture;

119 addto name also rrdropshadowpic;

120 rrdropshadowpic := image(fill rrtoplftcorner--rrtoprtcorner--

121 rrbotrtcorner--rrbotlftcorner--cycle withcolor

122 rrinnercolor);

123 addto name also rrdropshadowpic;

124 % name := image(fill rrtoplftcorner--rrtoprtcorner--

125 % rrbotrtcorner--rrbotlftcorner--cycle withcolor

126 % rrinnercolor);

127 picture rrtmpborder;

128 border := currentpicture;

129 if (rrnotop = false):

130 rrtmpborder := image(draw rrtopborder withcolor

131 rrtopbordercolor withpen rrtopborderpen);

132 addto border also rrtmpborder;

133 fi

134 if (rrnobot = false):

135 rrtmpborder := image(draw rrbotborder withcolor

136 rrbotbordercolor withpen rrbotborderpen);

137 addto border also rrtmpborder;

138 fi

139 if (rrnolft = false):

140 rrtmpborder := image(draw rrlftborder withcolor

141 rrlftbordercolor withpen rrlftborderpen);

142 addto border also rrtmpborder;

143 fi

144 if (rrnort = false):

145 rrtmpborder := image(draw rrrtborder withcolor

146 rrrtbordercolor withpen rrrtborderpen);

147 addto border also rrtmpborder;

148 fi

149 addto name also border;

150 rrtext :=

151 image(label(TEX("\parbox{"&decimal(rrtextwd)&"bp}{"&rrtextalign&rrtextfont&" "&rrbodytext&"}"),(0,0)) withcolor rrtextcolor;);

152 addto name also rrtext;

153 enddef;

